
Introduction

While a computer language's lexical grammar can usually be described using regular expressions, the
phrase structure of most languages cannot be expressed using regular grammars alone. Context-free
grammars are more powerful than regular grammars, and in fact suffice to describe most comput-
er language constructs. In addition, many different algorithms are available for generating efficient
parsers from various subsets of the context-free grammars.

This library provides a collection of objects for representing context-free grammars (CFGs), along
with methods for generating parsers from these grammars. Unlike most other parser generation sys-
tems, this library attempts to maximize generation speed so that the parser can be generated dynami-
cally, making it easier for grammar definitions to be completely integrated into the surrounding pro-
gram (as with the system described in [4] and [5]).

〈Module Dylan-user 1〉≡
define library grammar
  use common-dylan;

  〈Exports for the grammar library 1〉
end library;
          

〈Utility module definitions for the grammar library 1〉
〈Module definitions for the grammar library 4〉

Directed Graph Algorithms

Before we delve into the grammars themselves, first we'll introduce some directed graph (digraph)
algorithms that will be useful for several different grammar-related computations.

〈Exports for the grammar library1〉≡
export digraph;
          

This section continues on pages 4, 14 and 37.

This section is referenced on page 1.

〈Utility module definitions for the grammar library1〉≡
define module digraph
  use common-dylan;
  export find-rooted-scc;
end module;
          

This section continues on page 37.

This section is referenced on page 1.

Finding Strongly Connected Components of a Digraph

Imagine that we have a directed graph, and we want to compute some function on each of the nodes,
where the value of the function for a non-leaf node depends on the value of the function on some or
all of the node's children. If the graph were a tree or a directed acyclic graph (DAG), then we could
compute the function for all nodes using a depth-first traversal:

1. For leaf nodes compute the function using the basis rule.
2. For non-leaf nodes,

i. Recursively compute the function for each of the node's children, then
ii. Compute the value for the node itself.

If the graph has any cycles, the depth-first traversal will loop indefinitely. If we were to prevent infinite
looping by marking visited nodes and only visiting un-marked children, then we would compute the
function incorrectly for some of the nodes in the cycle, since we would fail to account for the circular
dependency.

What we need is an algorithm that will identify these cycles in the graph and propagate values to all
of the nodes in the cycle. An efficient algorithm to do this was devised by Eve and Kurki-Suonio [9],
who based it on an earlier algorithm by Robert Tarjan (FIXME cite references). Their algorithm is
explained in further detail in [8].

Eve and Kurki-Suonio's algorithm identifies strongly connected components (SCCs) in the graph,
that is, subsets of the graph where every node is reachable from every other. Once we have identified
SCCs, we can treat them as single nodes.

find-rooted-scc [Function]   

Find strongly-connected components in a directed graph

Signature:
find-rooted-scc root visit-function collapse-function ⇒

Arguments: ■ root
An instance of <object>.

■ visit-function
An instance of <function>.

■ collapse-function
An instance of <function>.

Values: None.
Description: The find-rooted-scc function implements Eve and Kurki-Suonio's algorithm for finding strong-

ly-connected components of a directed graph, under the assumption that all graph nodes of interest are
reachable from the root. Users of the function parameterize it using two function arguments. The
visit-function is called to visit a node; it is passed the node to be visited and a function that it
should call in turn to traverse the node's children. The collapse-function gets called whenever
a non-trivial (more than one node) strongly connected component is found; it is passed the head of
the SCC and the new node to be collapsed into the SCC.
We start with a stack, initially empty, and an array index, initially containing 0 for every node.

〈Module digraph 2〉≡
define method find-rooted-scc
    (root :: <object>,
     visit-function :: <function>,
     collapse-function :: <function>) => ();

  let stack :: <list> = #();
  let stack-depth :: <integer> = 0;
  let index :: <object-table> = make(<object-table>);

  〈Define the local traverse method 3〉
  traverse(root);
end method;
            

When we start to traverse a node, we push it onto the stack, and set its index to the current stack
depth.

〈Define the local traverse method3〉≡
local
  method traverse (node :: <object>) => ();
    stack := add!(stack, node);
    stack-depth := stack-depth + 1;
    let initial-stack-depth = stack-depth;
    index[node] := initial-stack-depth;

    〈Visit the node 3〉

    〈Determine whether we've found an SCC, and mark it if we have 4〉
  end method;
            

This section is referenced on page 3.

To visit the node, we call the user-supplied visit-function, passing it node and a function
that it can use to traverse the children of node. The (anonymous) function is called for each child to
recursively traverse it, and set the index of node to the minimum of the indices of children.

We treat an index of #f as meaning infinity.

〈Visit the node3〉≡
visit-function(node,
               method(child :: <object>) => ();
                   if (element(index, child, default: 0) = 0)
                     traverse(child);
                   end if;

                   let child-index = index[child];
                   let node-index = index[node];
                   if (child-index)
                     if (node-index)
                       index[node] := min(node-index, child-index);
                     else
                       index[node] := child-index;
                     end if;
                   end if;
               end method);
            

This section is referenced on page 3.

Now we're done traversing sub-nodes. If the node's index was reduced (because of the min), we know
it is because one of its descendents is also one of its ancestors (traversed at a lower stack depth), and is
therefore part of a cycle. In that case, we do nothing, leaving the node on the stack until the recursion
unwinds far enough to reach the head of the SCC.

If the index was not reduced, then node is the head node of an SCC (possibly a trivial one containing
only the node itself). When we find the head, we pop all of the members in the cycle from the stack,
call the user-supplied collapse-function, and mark each node with an index of infinity (#f)
so that it does not affect the computation of other SCCs.

〈Determine whether we've found an SCC, and mark it if we have4〉≡
if (index[node] = initial-stack-depth)
  while (first(stack) ~== node)
    collapse-function(node, first(stack));
          
    index[first(stack)] := #f;
    stack := tail(stack);
    stack-depth := stack-depth - 1;
  end while;

  index[node] := #f;
  stack := tail(stack);
  stack-depth := stack-depth - 1;
end if;
            

This section is referenced on page 3.

Representing Context-Free Grammars

The grammar module provides data structures and methods for representing context-free grammars
(CFGs), and for computing functions on elements of the grammar.

〈Exports for the grammar library 1〉+≡
export grammar;
          

〈Module definitions for the grammar library4〉≡
define module grammar
  use common-dylan;

  〈Modules imported by the grammar module 12〉

  〈Names exported by the grammar module 5〉
end module;
          

This section continues on page 14.

This section is referenced on page 1.

The Defining Characteristics of Context-Free Grammars

Formally, a context-free grammar is defined by the following characteristics:

■ A set of terminals, or tokens. Terminals are normally defined using a lexical grammar; they are
called terminals because they are indivisible as far as the CFG is concerned.

■ A set of nonterminals. While terminals represent single words in the grammar, nonterminals can
represent entire phrases. Nonterminals are defined in terms of terminals, other nonterminals, or
recursively in terms of themselves.

■ A start symbol, the nonterminal within a CFG that defines the phrase structure of the language
defined by the grammar.

■ A set of productions, each defining a possible expansion for a nonterminal in the grammar. The
left-hand side of a production is the nonterminal, and the right-hand side of a production is a se-
quence of zero or more terminals and nonterminals.

If we are using the grammar to define a parser, we may also need to associate an action with each
production. This action will be performed whenever the parser recognizes the production.

It's not really necessary to explicitly represent all of these characteristics in our <grammar> class,
if we assume that the set of nonterminals is exactly the set of all of the symbols that appear on the
left-hand side of productions in the grammar. Any symbol that is not a nonterminal can be assumed
to be a terminal.

We can do without the start symbol, too, if we pass it as an argument to every function that needs to
know. This allows us to easily use the same set of productions to define multiple grammar subsets.

We won't be defining a specific class to represent grammar symbols. Dylan's built-in <symbol>
class is a good choice, but any class will do, as long as we can compare grammar symbols for equality
using ==.

The <grammar> class

〈Names exported by the grammar module5〉≡
export <grammar>,
       grammar-symbol-productions,
       grammar-symbol-nonterminal?,
       grammar-symbol-nullable?,
       grammar-symbol-left-recursive?,
       grammar-symbol-first-symbols;
           

This section continues on page 8.

This section is referenced on page 4.

Though we're making <grammar> an abstract class, later we'll override make to instantiate a default
concrete subclass.

<grammar> [Open Abstract Instantiable Class]   

The class of context-free grammars.

Superclasses: <object>
Init-keywords: The constructors for subclasses of <grammar> should take a productions: keyword argument,

giving a sequence containing the productions that make up the grammar.
Definition: 〈Module grammar 5〉≡

define open abstract class <grammar> (<object>)
  // No additional slots
end class;
                  

This module continues on pages 6, 6, 7, 7, 8, 8, 9, 9, 9, 10, 10, 10, 11,
11, 11 and 12.

We need to be able to retrieve the productions associated with a given nonterminal (that is, the pro-
ductions in the grammar that have the given symbol as a left-hand side part).

grammar-symbol-productions [Open Generic Function]   

Returns the productions associated with the specified nonterminal symbol.

Signature:
grammar-symbol-productions grammar nonterminal-symbol ⇒ productions

Arguments: ■ grammar
An instance of <grammar>

■ nonterminal-symbol
An instance of <object>

Values: ■ productions
An instance of <sequence>

Description: Returns a sequence containing all of the productions in grammar with the specified nonter-
minal-symbol on their left-hand sides. Note that the sequence returned by grammar-sym-
bol-productions should not be modified.

Definition: 〈Module grammar5〉+≡
define open generic grammar-symbol-productions
    (grammar :: <grammar>,
     nonterminal-symbol :: <object>)
 => (productions :: <sequence>);
                  

grammar-symbol-nonterminal? [Open Generic Function]   

Tests whether a grammar symbol is a nonterminal.

Signature:
grammar-symbol-nonterminal? grammar symbol ⇒ nonterminal?

Arguments: ■ grammar
An instance of <grammar>

■ symbol
An instance of <object>

Values: ■ nonterminal?
An instance of <boolean>

Description: Returns a true value if the specified symbol is considered a nonterminal in grammar, or #f if the
symbol is a terminal symbol or unknown.

Definition: 〈Module grammar5〉+≡
define open generic grammar-symbol-nonterminal?
    (grammar :: <grammar>,
     symbol :: <object>)
 => (nonterminal? :: <boolean>);
                  

grammar-symbol-nullable? [Open Generic Function]   

Tests whether a given grammar symbol can derive the empty string.

Signature:
grammar-symbol-nullable? grammar symbol ⇒ nullable?

Arguments: ■ grammar
An instance of <grammar>

■ symbol
An instance of <object>

Values: ■ nullable?
An instance of <boolean>

Description: Returns a true value if the given symbol is a nonterminal symbol in grammar that can derive the
empty string (either directly or indirectly), or #f it is either a terminal symbol or a nonterminal that
is not nullable.

Definition: 〈Module grammar5〉+≡
define open generic grammar-symbol-nullable?
    (grammar :: <grammar>,
     symbol :: <object>)
 => (nullable? :: <boolean>);
                  

grammar-symbol-left-recursive? [Open Generic Function]   

Tests whether a given grammar symbol is defined in a left-recursive manner.

Signature:
grammar-symbol-left-recursive? grammar symbol ⇒ left-recursive?

Arguments: ■ grammar
An instance of <grammar>

■ symbol
An instance of <object>

Values: ■ left-recursive?
An instance of <boolean>

Description: Returns a true value if the given symbol is a nonterminal symbol in grammar that is defined left-
recursively (either directly or indirectly), or #f it is either a terminal symbol or a nonterminal that
does not have a left-recursive definition.

Definition: 〈Module grammar5〉+≡
define open generic grammar-symbol-left-recursive?
    (grammar :: <grammar>,
     symbol :: <object>)
 => (left-recursive? :: <boolean>);
                  

grammar-symbol-first-symbols [Open Generic Function]   

Returns the FIRST set of a grammar symbol.

Signature:
grammar-symbol-first-symbols grammar symbol ⇒ first-symbols

Arguments: ■ grammar
An instance of <grammar>

■ symbol
An instance of <object>

Values: ■ first-symbols

An instance of <sequence>
Description: Returns as a sequence the FIRST set of the specified grammar symbol, which consists of all of the

terminal symbols that could appear at the beginning of a nonempty string derived from the symbol.
The FIRST set for a terminal symbol consists only of the symbol itself.

Definition: 〈Module grammar5〉+≡
define open generic grammar-symbol-first-symbols
    (grammar :: <grammar>,
     symbol :: <object>)
 => (first-symbols :: <sequence>);
                  

The <production> class

〈Names exported by the grammar module 5〉+≡
export <production>, production-nonterminal, production-derives;
           

<production> [Open Instantiable Primary Class]   

Class for representing productions in context-free grammars.

Superclasses: <object>
Init-keywords: The class <production> and its subclasses support the following init-keywords:

■ nonterminal:
An instance of <object>.

This argument specifies the nonterminal symbol on the left-hand side of the grammar production.
This argument is required.

■ derives:
An instance of <sequence>.

This argument specifies a sequence of zero or more grammar symbols on the right-hand side of the
grammar production. This argument is required.

Description: A production consists of a nonterminal (the left-hand side) and a sequence of grammar symbols (the
right-hand side). The right-hand side, representing a sequence of symbols that the nonterminal can
derive, is stored as a vector to maintain quick access to individual symbols.

Definition: 〈Module grammar5〉+≡
define open primary class <production> (<object>)
  constant slot production-nonterminal :: <object>,
    required-init-keyword: nonterminal:;
  constant slot production-derives :: <vector>,
    required-init-keyword: derives:;
end class;
                  

A Concrete Implementation of the Grammar Representation Classes

We won't be exporting the name of the <concrete-grammar> class. We will, however, overload
the constructor for <grammar> so that it constructs a <concrete-grammar> instance.

The grammar-nonterminals slot contains a table mapping nonterminal grammar symbols to
structures containing information about each nonterminal.

〈Module grammar5〉+≡
define class <concrete-grammar> (<grammar>)
  constant slot grammar-nonterminals :: <object-table>
    = make(<object-table>);
end class;
            
define sealed method make
    (class == <grammar>, #key productions :: <sequence>)
 => (object :: <concrete-grammar>);
  make(<concrete-grammar>, productions: productions);
end method; 
            

Elements in the grammar-nonterminals table for a grammar are instances of <nontermi-
nal>. This object collects the productions that the nonterminal may derive, and memoizes the result
of determining the FIRST set for the nonterminal.

〈Module grammar5〉+≡
define class <nonterminal> (<object>)
  slot nonterminal-productions :: <list> = #();
  slot nonterminal-nullable? :: <boolean> = #f;
  slot nonterminal-left-recursive? :: <boolean> = #f;
  slot nonterminal-first-symbols :: <list>;
end class;
            

When we initialize a <concrete-grammar> instance, we collect productions for a given nonter-
minal symbol into its corresponding <nonterminal> object.

If a production can derive the empty string trivially (because its production-derives is empty),
then we can go ahead and mark its nonterminal as nullable. Productions that are nullable because all
of the symbols on their right-hand sides are nullable will be found by compute-nullable-non-
terminals.

〈Module grammar5〉+≡
define sealed method initialize
    (instance :: <concrete-grammar>,
     #key productions :: <sequence>, #all-keys)
 => ();
  next-method();
  let nonterminals = instance.grammar-nonterminals;

  for (production in productions)
    let nonterminal
       = element(nonterminals, production.production-nonterminal, default: #f)
         | (nonterminals[production.production-nonterminal]
              := make(<nonterminal>));

    nonterminal.nonterminal-productions
      := add!(nonterminal.nonterminal-productions, production);

    if (empty?(production.production-derives))
      nonterminal.nonterminal-nullable? := #t;
    end if;



10

  end for;

  compute-nullable-nonterminals!(instance);
end method;
            

Given this, we can implement the methods on <grammar> that we mentioned above.

〈Module grammar5〉+≡
define method grammar-symbol-productions
    (grammar :: <concrete-grammar>,
     nonterminal-symbol :: <object>)
 => (productions :: <sequence>);
  let nonterminal = element(grammar.grammar-nonterminals,
                            nonterminal-symbol,
                            default: #f);
  if (nonterminal)
     nonterminal.nonterminal-productions
  else
     error("asked to find productions for terminal symbol %=",
           nonterminal-symbol);
  end if;
end method;
            

〈Module grammar5〉+≡
define method grammar-symbol-nonterminal?
    (grammar :: <concrete-grammar>, symbol :: <object>)
 => (nonterminal? :: <boolean>);
  if (element(grammar.grammar-nonterminals, symbol, default: #f))
     #t
  else
     #f
  end if;
end method;
            

If a symbol is a terminal symbol, then it is obviously not nullable. The nullibility of nonterminals is
computed during the initialization for the grammar, as we'll see.

〈Module grammar5〉+≡
define method grammar-symbol-nullable?
    (grammar :: <concrete-grammar>, symbol :: <object>)
 => (nullable? :: <boolean>);
  let nonterminal = element(grammar.grammar-nonterminals,
                            symbol,
                            default: #f);
  if (nonterminal)
     nonterminal.nonterminal-nullable?
  else
     #f
  end if;
end method;
            

The FIRST set of a terminal contains itself only. For a nonterminal, we first check to see if its
FIRST set has already been computed (either directly or in the process of computing another symbol's
FIRST set). If not we call compute-first-symbols!, which sets the nonterminal's nonter-
minal-first-symbols field.

〈Module grammar5〉+≡
define method grammar-symbol-first-symbols
    (grammar :: <concrete-grammar>, symbol :: <object>)
 => (first-symbols :: <sequence>);
  let nonterminal :: false-or(<nonterminal>)
    = element(grammar.grammar-nonterminals, symbol,
              default: #f);
  if (nonterminal)
    if (~slot-initialized?(nonterminal, nonterminal-first-symbols))
      compute-first-symbols!(grammar, nonterminal);
    end if;
    nonterminal.nonterminal-first-symbols;
  else
    list(symbol);
  end if;
end method;
            

The information required by grammar-symbol-left-recursive? is also computed by com-
pute-first-symbols!.

〈Module grammar5〉+≡
define method grammar-symbol-left-recursive?
    (grammar :: <concrete-grammar>, symbol :: <object>)
 => (left-recursive? :: <boolean>);
  let nonterminal :: false-or(<nonterminal>)
    = element(grammar.grammar-nonterminals, symbol,
              default: #f);
  if (nonterminal)
    if (~slot-initialized?(nonterminal, nonterminal-first-symbols))
      compute-first-symbols!(grammar, nonterminal);
    end if;
    nonterminal.nonterminal-left-recursive?
  else
    #f
  end if
end method;
            

Computing Nullibility for Grammar Nonterminals

To determine which symbols are nullable, we will use a straightforward fixed-point iteration algo-
rithm. (See, for instance, [2] or [1].)

〈Module grammar5〉+≡
define method compute-nullable-nonterminals!
    (grammar :: <concrete-grammar>) => ();
  let nonterminals :: <object-table> = grammar.grammar-nonterminals;
  let changed? :: <boolean> = #t;

  while (changed?)
    changed? := #f;
    for (nonterminal in nonterminals)

      〈Compute nullibility for nonterminal and set changed? when appropriate 12〉

    end for;
  end while;
end method;
            

A nonterminal is nullable if any of the productions it derives consists only of nullable symbols.

〈Compute nullibility for nonterminal and set changed? when appropriate12〉≡
if (~nonterminal.nonterminal-nullable?)
  if (any?(method(production)
             every?(curry(grammar-symbol-nullable?, grammar),
                    production.production-derives);
           end method,
           nonterminal.nonterminal-productions))
    nonterminal.nonterminal-nullable? := #t;
    changed? := #t;
  end if;
end if;
            

This section is referenced on page 11.

Computing FIRST Sets for Grammar Nonterminals

We could compute the FIRST sets for all of the nonterminals at initialization time, as part of the same
fixed-point iteration that we used to compute whether they were nullable. However, this involves
doing many set union operations repeatedly to see if the result has changed between iterations. We
can do much better than that.

The key is the find-rooted-scc function we described earlier. It only computes values for a
node once, and handles cycles in the graph (caused by left dependence in the grammar productions)
without looping, propagating values to nodes that need to be recomputed when the cycle is found.

〈Modules imported by the grammar module12〉≡
use digraph;
            

This section is referenced on page 4.

〈Module grammar5〉+≡
define method compute-first-symbols!
    (grammar :: <concrete-grammar>,
     nonterminal :: <nonterminal>)
  => ();
  local

    〈Define local visit and collapse methods for compute-first-symbols! 13〉
  find-rooted-scc(nonterminal, visit, collapse);
end method;
            

To compute the FIRST set of a nonterminal, we union the FIRST sets of the symbols on which it has a
left-dependency. That is, for each production, we look at the FIRST set of the first symbol, unioning
it with the second if the first was nullable, and with the third if the second was nullable, and so on
up through the first non-nullable symbol.

We initialize the nonterminal's FIRST set to the empty set so that recursive calls will have a starting
point.

〈Define local visit and collapse methods for compute-first-symbols!13〉≡
method visit
    (nonterminal :: <nonterminal>,
     traverse :: <function>)
 => ();
  nonterminal.nonterminal-first-symbols := #();

  for (production in nonterminal.nonterminal-productions,
       first-symbols = #()
         then for (symbol in production.production-derives,
                   production-first-symbols = first-symbols
                     then
                       begin
                         let child :: false-or(<nonterminal>)
                           = element(grammar.grammar-nonterminals,
                                     symbol, default: #f);
                         if (child)
                           if (~slot-initialized?(child,
                                                  nonterminal-first-symbols))
                             traverse(child);
                           elseif (child == nonterminal)
                             nonterminal.nonterminal-left-recursive? := #t;
                           end if;
                           union(child.nonterminal-first-symbols,
                                 production-first-symbols);
                         else
                           add-new(production-first-symbols, symbol);
                         end if;
                       end,
                   prefix? = #t
                     then grammar-symbol-nullable?(grammar, symbol),
                   while: prefix?)
              finally
                production-first-symbols;
              end for)
  finally
    nonterminal.nonterminal-first-symbols := first-symbols;
  end for;
end method,
            

This section continues on page 13.

This section is referenced on page 12.

Mutually left-recursive grammar symbols share the same FIRST sets. When we collapse them into a
single SCC, we propagate the FIRST set from the SCC head (which has a correctly computed set) to
the other elements (which may not). We also set the nonterminal-left-recursive? slot.

〈Define local visit and collapse methods for compute-first-symbols! 13〉+≡
method collapse
    (scc-head :: <nonterminal>,
     scc-other :: <nonterminal>)
 => ();
  scc-other.nonterminal-first-symbols
    := scc-head.nonterminal-first-symbols;
  scc-other.nonterminal-left-recursive?
    := scc-head.nonterminal-left-recursive?

    := #t;
end method;
            

Parser Automata

A regular language (described by a regular expression/regular grammar) can be parsed using a simple
deterministic finite-state automaton. A parser for a general context-free grammar can be built using
a finite-state automaton and a stack to keep track of the current parse state. (This is called a push-
down automaton.) In this section, we'll develop the algorithms for building the finite-state automaton
needed for parsing context-free grammars.

〈Exports for the grammar library 1〉+≡
export parser-automaton;
          

〈Module definitions for the grammar library 4〉+≡
define module parser-automaton
  use common-dylan;
  use grammar;

  〈Modules imported by the parser-automaton module 38〉

  〈Names exported by the parser-automaton module 15〉
end module;
          

Our goal is to be able to instantiate a <parser> object, given a <grammar> that it should recognize.
If we were to go directly from a <grammar> to a <parser>, though, then we would end up re-
building the state machine every time we instantiated a parser for the same grammar. So, instead, we
will split up the interface for the generation of parsers into two stages:

1. The client instantiates a <parser-automaton> object from a <grammar> object and a start
symbol.

2. Each time a parser object is needed, the client instantiates a <parser> object from the <pars-
er-automaton> to do the actual parse.

Classes of Automata

A grammar defines a language, the set of strings of terminal symbols that can be derived from a
grammar by starting with the start symbol, and replacing nonterminals with right-hand sides of pro-
ductions they derive, until no unexpanded nonterminals remain. We could do this in any order; if
we always replace the leftmost nonterminal, it is called a leftmost derivation; if we always pick the
nonterminal that is the furthest to the right then it is called a rightmost derivation.

While a derivation tells how to go from a grammar to a string in the language, a parse goes the other
way, determining what derivations can be used to derive the string. Performing a parse results in a
parse tree, which identifies which productions were chosen to substitute for each nonterminal in order
to derive the input string.

One way of doing this would be to do a search with backtracking, but this is time-consuming and
inefficient. Aside from backtracking, there are two primary generally-used approaches to parsing:

■ Top-down parsing starts from the start symbol and works down, guessing at each step which
production was used to derive a substring by looking at its beginning, and comparing it with the
possible first symbols of the available productions. Top-down parsing is also called predictive
parsing (because it predicts at the beginning of the substring which production should be chosen)
and LL parsing (since it reads the input left-to-right, and corresponds to the leftmost derivation).

■ Bottom-up parsing works up from the bottom, discovering when it has seen the end of a production
(and maybe a few symbols of lookahead) that it has just matched a nonterminal. The parse is done
when it discovers it has matched the start symbol. Bottom-up parsing is also called LR parsing,
because it reads the input left-to-right, and corresponds to the rightmost derivation.

Variations on top-down parsing include:

■ LL(1) parsing, where one symbol of lookahead is used to determine which production applies.
Most practical programming languages can be recognized using an LL(1) parser. In addition, re-
cursive-descent LL(1) parsers are easy to build by hand, and they are easy to debug. However,
sometimes the transformations necessary to make a grammar map to an LL(1) parser render it dif-
ficult to read.

■ LL(k) parsing, where k symbols of lookahead (k > 1) are used to determine which production
applies. This is more powerful than LL(1) techniques, though it is more difficult to build the parser
by hand. LL(k) parsers are also easy to debug.

Variations on bottom-up parsing include:

■ LR(0) parsing, where no symbols of lookahead are required. LR(0) parsing is not powerful enough
to parse practical programming languages.

■ SLR(1) (simple LR) parsing, where one symbol of lookahead is used. SLR(1) parsing is based
on the addition of coarse-grained lookahead information to part of the LR(0) parser automaton
(helping it decide which nonterminal has just been recognized), and the class of grammars that an
SLR(1) automaton can recognize is a superset of the LR(0) grammars. SLR(1) parsers are powerful
enough to parse many practical grammars, but some common constructs are beyond their reach.

■ LALR(1) parsing also uses one symbol of lookahead to decide which production has just been
recognized, but its lookahead sets are finer-grained, including only the symbols that could appear
after the production, in the current context. LALR(1) parsing is powerful enough to recognize
almost all practical programming languages, and is very commonly used.

■ LR(1) parsing is the most powerful of the methods requiring one symbol of lookahead. An LR(1)
automaton is similar to a LALR(1) automaton, except that some of the states of the deterministic
finite automaton are split up in order to make the lookahead sets even finer grained. This can make
the automaton unreasonably large, for only a marginal gain in power over a LALR(1) parser.

■ LR(k) parsing, where k symbols (k > 1) of lookahead are used. Algorithms requiring more than one
symbol of lookahead are seldom used for LR parsing, because they require large parser automata.

Subclasses of our <parser-automaton> can potentially implement any of these classes of au-
tomata.

Interface

〈Names exported by the parser-automaton module15〉≡
export <parser-automaton>, make-parser-automaton;
            

This section continues on pages 17, 18 and 21.

This section is referenced on page 14.

Just as we did with the <grammar> class, later we'll override the make method to instantiate concrete
subclasses.

<parser-automaton> [Open Abstract Instantiable Class]   

The class of automata used for parsing context-free grammars.

Superclasses: <object>
Init-keywords: Subclasses of <parser-automaton> support the following init-keywords:

■ grammar:
An instance of <grammar> specifying the grammar that the automaton should recognize. If the
argument is not supplied, or the supplied argument is not a general instance of <grammar>, then
an error is signaled.

■ start-symbols:
An instance of <sequence> specifying which of the grammar's nonterminal symbols should be
considered as start symbols.

■ end-symbol:
An instance of <object> specifying the terminal symbol that the parser should treat as indicating
the end of the input.

■ class:
An instance of <symbol> specifying the class of recognizer to be built. Currently supported
classes include #"LR-0" for an LR(0) automaton, #"SLR-1" for an SLR(1) automaton, and
#"LALR-1" for a LALR(1) automaton. The default value is #"LALR-1".

■ full-lookahead?:
An instance of <object>. If this argument has a boolean true value then for lookahead-based
automata, lookaheads will be computed even when not strictly necessary.

Description: Encapsulates the finite-state automaton portion of a parser for a context-free grammar. This class
is intended to be used to factor out the portion of a parser that is constant and is time-intensive to
construct, to reduce the time required to instantiate multiple parsers that recognize the same grammar.

Definition: 〈Module parser-automaton 16〉≡
define open abstract class <parser-automaton> (<object>)
  // No additional slots
end class;
                

This module continues on pages 16, 17, 17, 18, 19, 20, 20, 21, 21, 22, 22, 23, 23,
23, 24, 24, 24, 24, 25, 25, 25, 25, 26, 26, 26, 26, 27, 27, 31, 31,
32, 32, 32, 33, 33, 34, 34, 34, 35, 35, 36, 40, 41, 42, 42, 42 and 43.

make-parser-automaton [Open Generic Function]   

Factory method for instantiating subclasses of <parser-automaton>

Definition The <parser-automaton> class is an abstract instantiable class, with a make method defined
as follows:

〈Module parser-automaton16〉+≡
define sealed method make
    (type == <parser-automaton>,
     #rest key-value-pairs,

     #key class :: <symbol> = #"LALR-1",
     #all-keys)
 => (instance :: <parser-automaton>);
  apply(make-parser-automaton, class, key-value-pairs);
end method;
                

The make-parser-automaton generic function serves as a factory method for instantiating a
concrete subclass of <parser-automaton>.

〈Module parser-automaton16〉+≡
define open generic make-parser-automaton
    (grammar-class :: <symbol>, #rest key-value-pairs, #key)
 => (instance :: <parser-automaton>);
                

Error Conditions

Subclasses of the error condition class <parser-automaton-error> may be signaled when an
error is detected during automaton construction.

〈Names exported by the parser-automaton module 15〉+≡
export
  <parser-automaton-error>,
  parser-automaton-error-productions,
  parser-automaton-error-inputs;
            

<parser-automaton-error> [Open Abstract Class]   

The class of error conditions signaled during parser automaton construction.

Superclasses: <error>
Init-keywords: Subclasses of <parser-automaton-error> accept the following init-keywords:

■ productions:
An instance of <sequence>. A sequence containing the <production> objects that may con-
flict to cause the error. This keyword is required.

■ inputs:
An instance of <sequence>. The input symbols, if any, where the conflict might occur. Defaults
to an empty sequence.

Operations: ■ parser-automaton-error-productions
Returns an instance of <sequence> containing the <production> objects that may conflict
to cause the error.

■ parser-automaton-error-inputs
Returns the input symbols, if any, where the conflict might occur.

Definition 〈Module parser-automaton16〉+≡
define open abstract class <parser-automaton-error> (<error>)
  constant slot parser-automaton-error-productions :: <sequence>,
    required-init-keyword: productions:;
  constant slot parser-automaton-error-inputs :: <sequence> = #(),
    init-keyword: inputs:;

end class;
                

Parser Automata for Bottom-Up Parsing

An LR (bottom-up) parser is, as we mentioned earlier, a pushdown automaton. In particular, it contains
the following elements:

■ An input source, which the parser can read one symbol at a time.

■ A stack, used to hold grammar symbols in preparation for figuring out what production they belong
to, as well as the automaton state when the symbols were read or recognized.

■ A finite-state automaton with edges labeled with grammar symbols. The automaton's state is the
state on the top of the stack. On seeing a symbol, the automaton can perform one of the following
actions:

● Read a new terminal (token) symbol from the input and push it on to the top of stack, along with
a specified new state. This is called a shift action.

● For some specified grammar production, pop as many elements off the stack as the production
has symbols on the right-hand side. Then, from the state that is now on top of the stack, push the
nonterminal on the left-hand side of the production, along with the state specified by the current
state's goto transition on that symbol. This is called a reduce action.

● Stop the parse, and report that the input matches the grammar. This is called an accept action.

● Stop the parse and signal an error.
At each reduce operation, it replaces the symbols on the top of the stack with the nonterminal whose
production they match, implicitly building a parse tree.

Our task now is to build this automaton from the grammar.

Interface

〈Names exported by the parser-automaton module 15〉+≡
export <lr-parser-automaton>;
export lr-parser-automaton-initial-state,
       lr-parser-automaton-terminal-transition,
       lr-parser-automaton-nonterminal-transition,
       lr-parser-automaton-transition-terminals,
       lr-parser-automaton-transition-nonterminals;
            

<lr-parser-automaton> [Open Abstract Instantiable Class]   

The class of automata used for bottom-up parsing of context-free grammars.

Superclasses: <parser-automaton>
Init-keywords: Same as with <parser-automaton>.
Description: Encapsulates the state machine automaton portion of a bottom-up (LR) parser.
Definition: 〈Module parser-automaton16〉+≡

define open abstract class <lr-parser-automaton> (<parser-automaton>)

  // No additional slots
end class;
                

The following functions allow bottom-up parsers to access the content of an LR automaton, indepen-
dent of its internal representation.

lr-parser-automaton-initial-state [Open Generic Function]   

Returns the initial state of a bottom-up parser automaton.

Signature:
lr-parser-automaton-initial-state automaton start-symbol ⇒ initial-state

Arguments: ■ automaton
An instance of <lr-parser-automaton>.

■ start-symbol
An instance of <object>.

Values: ■ initial-state
An instance of <object>

Description: Returns the initial state of the given parser automaton for the given start symbol. The returned state
should be considered an implementation-dependent opaque object.

Definition: 〈Module parser-automaton16〉+≡
define open generic lr-parser-automaton-initial-state
    (automaton :: <lr-parser-automaton>, start-symbol :: <object>)
 => (state :: <object>);
                

lr-parser-automaton-terminal-transition [Open Generic Function]   

Parser action function for terminal transitions.

Signature:
lr-parser-automaton-terminal-transition automaton state terminal ⇒ action data

Arguments: ■ automaton
An instance of <lr-parser-automaton>

■ state
An instance of <object>

■ terminal
An instance of <object>

Values: ■ action
One of the symbols #"shift", #"reduce", #"error", or #"accept".

■ data
An instance of <object>

Description: Returns the action that an LR parser should take on a given input. Possible action values are

■ #"shift"
Indicates that the parser should shift the current input symbol and transition to the new state returned
in data.

■ #"reduce"
Indicates that the parser should reduce the instance of <production> returned in data.

■ #"error"

Indicates that no transition on the given terminal is possible from the given state.
■ #"accept"

Indicates that parsing has successfully completed.
Definition: 〈Module parser-automaton16〉+≡

define open generic lr-parser-automaton-terminal-transition
    (automaton :: <lr-parser-automaton>,
     state :: <object>,
     terminal :: <object>)
 => (action :: one-of(#"shift", #"reduce", #"error", #"accept"),
     data :: <object>);
                

lr-parser-automaton-nonterminal-transition [Open Generic Function]   

Parser action function for nonterminal transitions.

Signature:
lr-parser-automaton-nonterminal-transition automaton state nonterminal ⇒
new-state

Arguments: ■ automaton
An instance of <lr-parser-automaton>

■ state
An instance of <object>

■ nonterminal
An instance of <object>

Values: ■ new-state
An instance of <object>

Description: Returns the new state to which the parser should transition on a nonterminal (after a reduction).
Definition: 〈Module parser-automaton16〉+≡

define open generic lr-parser-automaton-nonterminal-transition
    (automaton :: <lr-parser-automaton>,
     state :: <object>,
     nonterminal :: <object>)
 => (new-state :: <object>);
                

lr-parser-automaton-transition-terminals [Open Generic Function]   

Queries the parser action function for legal terminal transitions.

Signature:
lr-parser-automaton-transition-terminals automaton state ⇒ terminals

Arguments: ■ automaton
An instance of <lr-parser-automaton>

■ state
An instance of <object>

Values: ■ terminals
An instance of <sequence>, or #f.

Description: Returns a <sequence> containing the terminal symbols on which the parser automaton can tran-
sition in the given state, or #f if this information is unavailable for whatever reason.

Definition: 〈Module parser-automaton16〉+≡
define open generic lr-parser-automaton-transition-terminals
    (automaton :: <lr-parser-automaton>,
     state :: <object>)
 => (terminals :: false-or(<sequence>));
                

lr-parser-automaton-transition-nonterminals [Open Generic Function]   

Queries the parser action function for nonterminal transitions.

Signature:
lr-parser-automaton-transition-nonterminals automaton state ⇒ nonterminals

Arguments: ■ automaton
An instance of <lr-parser-automaton>

■ state
An instance of <object>

Values: ■ nonterminals
An instance of <sequence>, or #f.

Description: Returns a <sequence> containing the nonterminal symbols on which the parser automaton can
transition in the given state, or #f if this information is unavailable for whatever reason.

Definition: 〈Module parser-automaton16〉+≡
define open generic lr-parser-automaton-transition-nonterminals
    (automaton :: <lr-parser-automaton>,
     state :: <object>)
 => (nonterminals :: false-or(<sequence>));
                

Bottom-up Parser Generation Errors

Two things that might potentially go wrong and make it impossible to generate a bottom-up parser
automaton of a given class include:

■ A reduce/reduce conflict occurs when more than one production could be reduced in a state, and
there is not enough information to determine which one to reduce given some input.

■ A shift/reduce conflict occur when a state contains both shift transitions and possible reductions,
and there is not enough information to determine whether to shift or to reduce given some input.

The builders for each class of automata will check for these types of conflicts in the automata they
have constructed, and signal one of the following conditions if any conflicts are present.

〈Names exported by the parser-automaton module 15〉+≡
export
  <parser-automaton-reduce/reduce-error>,
  <parser-automaton-shift/reduce-error>,
  <parser-automaton-reduce/reduce-restart>,
  <parser-automaton-shift/reduce-restart>;
            

<parser-automaton-reduce/reduce-error> [Class]   

The class of error conditions signaled when a reduce/reduce conflict is detected.

Superclasses: <parser-automaton-error>
Init-keywords: ■ productions:

An instance of <sequence>. A sequence containing the <production> objects that may con-
flict to cause the error.

■ inputs:
An instance of <sequence>. The input symbols, if any, where the conflict might occur. Defaults
to an empty sequence.

Operations: ■ parser-automaton-error-productions
Returns an instance of <sequence> containing the <production> objects that may conflict
to cause the error.

■ parser-automaton-error-inputs
Returns the input symbol, if any, where the conflict might occur.

Definition: 〈Module parser-automaton16〉+≡
define class <parser-automaton-reduce/reduce-error>
    (<parser-automaton-error>)
  // No additional slots
end class;
                

<parser-automaton-shift/reduce-error> [Class]   

The class of error conditions signaled when a shift/reduce conflict is detected.

Superclasses: <parser-automaton-error>
Init-keywords: ■ productions:

An instance of <sequence>. A sequence containing the <production> objects that may con-
flict to cause the error. The first <production> object should be the production that would be
reduced, and the remaining productions should be productions that would cause a shift.

■ inputs:
An instance of <sequence>. The input symbols, if any, where the conflict might occur. Defaults
to an empty sequence.

Operations: ■ parser-automaton-error-productions
Returns an instance of <sequence> containing the <production> objects that may conflict
to cause the error.

■ parser-automaton-error-inputs
Returns the input symbol, if any, where the conflict might occur.

Definition: 〈Module parser-automaton16〉+≡
define class <parser-automaton-shift/reduce-error>
    (<parser-automaton-error>)
  // No additional slots
end class;
                

<parser-automaton-reduce/reduce-restart> [Class]   

The class of recovery restarts for recovering after a reduce/reduce conflict is detected.

Superclasses: <restart>
Init-keywords: ■ action:

An instance of <production>. The production that should be chosen in preference to the other
conflicting productions.

Definition: 〈Module parser-automaton16〉+≡
define class <parser-automaton-reduce/reduce-restart> (<restart>)
  constant slot parser-automaton-restart-action :: <production>,
    required-init-keyword: action:;
end class;
                

<parser-automaton-shift/reduce-restart> [Class]   

The class of recovery restarts for recovering after a shift/reduce conflict is detected.

Superclasses: <restart>
Init-keywords: ■ action:

The value #"shift", indicating that the input symbol should be shifted, or an instance of <pro-
duction>, indicating that the production should be reduced..

Definition: 〈Module parser-automaton16〉+≡
define class <parser-automaton-shift/reduce-restart> (<restart>)
  constant slot parser-automaton-restart-action :: 
    type-union(singleton(#"shift"), <production>),
    required-init-keyword: action:;
end class;
                

Generating the Canonical LR Automaton from a Grammar

To build a practical parser automaton such as an SLR(1) or LALR(1) automaton, we need to first
construct the canonical LR automaton, which corresponds to the LR(0) automaton for the grammar.
We will do this by making the canonical automaton a base class for the other bottom-up parser au-
tomaton classes, and making use of implementation inheritance.

The LR(0) automaton is generally not of much practical use in and of itself (but see [17]).

〈Module parser-automaton16〉+≡
define abstract class <canonical-lr-parser-automaton>
    (<lr-parser-automaton>)

  〈Slots for the <canonical-lr-parser-automaton> class 28〉
end class;
            
define method initialize
    (automaton :: <canonical-lr-parser-automaton>,
     #next next-method,
     #key grammar :: <grammar>,
          start-symbols :: <object>,
          end-symbol :: <object>,
     #all-keys)
 => ();
  local

    〈Local methods for the <canonical-lr-parser-automaton> class initialize method 29〉;
  next-method();

  〈Construct the start state and its successors 31〉
end method;
            

LR Items

The task of the automaton is to keep track of what productions we might be in the middle of parsing
at any given point, and how far along we are in parsing them.

The data structure we use to keep track of how far along we are is called an item or configuration. An
item consists of a grammar production and a marker called dot that tells us where we are in parsing
the production. The dot can be placed before any of the symbols in the production, or at the end.

〈Module parser-automaton16〉+≡
define abstract class <lr-item> (<object>)
  constant slot item-production :: <production>,
    required-init-keyword: production:;
  constant slot item-dot :: <integer>,
    required-init-keyword: dot:;
end class;
              

To instantiate <lr-item> objects, we will use the factory method specialized on the automaton
class.

〈Module parser-automaton16〉+≡
define generic make-lr-item
    (automaton :: <lr-parser-automaton>, #key, #all-keys)
 => (instance :: <lr-item>);
              

The first thing we need to know about an item is whether or not dot is at the end, indicating that the
production may be ready to reduce. Such items are called final.

〈Module parser-automaton16〉+≡
define method item-dot-at-end?
    (item :: <lr-item>)
 => (at-end? :: <boolean>);
  item.item-dot = item.item-production.production-derives.size;
end method;
              

We also want to know the symbol after dot. The symbols after the dots of items are the symbols we
will use to transition from one state in our automaton to the next.

〈Module parser-automaton16〉+≡
define method item-symbol-after-dot
    (item :: <lr-item>)
 => (symbol :: <object>);
  item.item-production.production-derives[item.item-dot];
end method;
              

Once we've transitioned on a symbol, then part of the new state corresponds to a new item with the
dot moved up by one.

〈Module parser-automaton16〉+≡
define method item-successor
    (item :: <lr-item>)
 => (new-item :: <lr-item>);
  make(object-class(item),
       production: item.item-production,
       dot: item.item-dot + 1);
end method;
              

Items are equal if they came from the same production and their dots are in the same place.

〈Module parser-automaton16〉+≡
define method item=? (item1 :: <lr-item>, item2 :: <lr-item>)
 => (result :: <boolean>);
  item1.item-production == item2.item-production
    & item1.item-dot = item2.item-dot;
end method;
              

LR Itemsets

Since during parsing we don't determine what production to reduce until we reach the end, at any point
we might be in the middle of recognizing several different possible productions. Therefore, states in
our state machine correspond not to individual items, but to itemsets (also called configuration sets).

Items in itemsets come from two sources. Items that are present because they are the successors of
items in a previous state are called kernel items. When an item in an itemset has the dot immediately
before a nonterminal, then in that context it becomes possible to start trying to recognize the produc-
tions that the nonterminal can derive. Adding these non-kernel items to the itemset (all of which have
their dots at the left-hand side) is called completing the itemset, and the items are called completion
items.

We'll keep the kernel items and completion items separate, since the kernel items are sufficient to
uniquely define an itemset. In fact, we don't need to store the completion items at all, since they are
only used while generating successor states.

〈Module parser-automaton16〉+≡
define abstract class <lr-itemset> (<object>)
  constant slot itemset-kernel-items :: <list>,
    required-init-keyword: kernel-items:;

  〈Additional slots for the <lr-itemset> class 29〉
end class;
              

Again, to instantiate <lr-itemset> objects, we will use a factory method specialized on the au-
tomaton class.

〈Module parser-automaton16〉+≡
define generic make-lr-itemset
    (automaton :: <lr-parser-automaton>, #key, #all-keys)
 => (instance :: <lr-itemset>);

Locating LR Itemsets

LR itemsets are generated as successors of other LR itemsets. When we generate a new itemset, we
need to check to see if we've already generated an identical state. The Dylan language provides a
user-extensible hashtable facility, allowing us to store already-generated states in a table, keyed on
the itemset's kernel.

To make use of this facility, we need to define a subclass of the <table> class.

〈Module parser-automaton16〉+≡
define class <lr-itemset-table> (<table>)
  // No additional slots
end class;
              

Then, we provide two functions: an equality test function and a hash computation function for keys.
For the equality function, we check that the sizes are identical, and that every item in the first set has
a corresponding item in the second.

〈Module parser-automaton16〉+≡
define function itemset-kernel=?
     (items1 :: <list>, items2 :: <list>)
  => (equal? :: <boolean>);
  (size(items1) = size(items2))
    & every?(method(item) member?(item, items2, test: item=?) end,
             items1);
end function;
              

The hash code of an individual item is computed using the object-hash codes of its production
and its dot.

〈Module parser-automaton16〉+≡
define method item-hash
    (item :: <lr-item>, initial-state :: <object>)
 => (hash-id :: <integer>, hash-state :: <object>);
  let (production-id, production-state)
    = object-hash(item.item-production, initial-state);
  let (dot-id, dot-state) = object-hash(item.item-dot, production-state);
  let merged-id = merge-hash-ids(production-id, dot-id, ordered: #t);
  values(merged-id, dot-state);
end method;
              

The hash function for the sequence of items computes the hash value for each individual item and
combines them using merge-hash-ids. Order in the sequence is not significant.

〈Module parser-automaton16〉+≡
define function itemset-kernel-hash
    (items :: <list>, initial-state :: <object>)
 => (hash-id :: <integer>, hash-state :: <object>);
  if (empty?(items))
    values(0, initial-state)
  else
    let (hash-id :: <integer>, hash-state :: <object>)

      = item-hash(first(items), initial-state);

    for (item in items, first? = #t then #f)
      if (~first?)
        let (item-hash-id :: <integer>, item-hash-state :: <object>)
          = item-hash(item, hash-state);
        hash-id := merge-hash-ids(hash-id, item-hash-id, ordered: #f);
        hash-state := item-hash-state;
      end if;
    end for;

    values(hash-id, hash-state);
  end if;
end function;
              

Now we have to provide a method for table-protocol specialized on our class. The method
should return the two functions we defined.

〈Module parser-automaton16〉+≡
define method table-protocol(table :: <lr-itemset-table>)
  => (test-function :: <function>, hash-function :: <function>);
  values(itemset-kernel=?, itemset-kernel-hash);
end method;
              
define sealed domain table-protocol(<lr-itemset-table>);
              

Completing LR Itemsets

Completion items can give rise to their own completion items, so we will build the completion item
set recursively.

〈Module parser-automaton16〉+≡
define method completion-items
    (automaton :: <lr-parser-automaton>,
     grammar :: <grammar>,
     kernel-items :: <sequence>)
 => (completion-items :: <sequence>);
  let completed-nonterminals :: <list> = #();
  let completion-items :: <list> = #();

  local

    〈Local methods for completion-items 28〉;

  scan-items(kernel-items);

  completion-items;
end method;
              

We loop through the unscanned items (initially the kernel items), collecting new completion items in
completion-items. When we find a nonterminal after dot in an item, and its productions haven't
been added yet, then we add them to the list, calling scan-items recursively to scan them in turn.

〈Local methods for completion-items28〉≡
method scan-items
    (unscanned-items :: <sequence>)
 => ();
  for (item :: <lr-item> in unscanned-items)
    if (~item-dot-at-end?(item))
      let symbol = item-symbol-after-dot(item);
      if (grammar-symbol-nonterminal?(grammar, symbol)
            & ~member?(symbol, completed-nonterminals))
        let added-completions
          = map(as-completion-item,
                grammar-symbol-productions(grammar, symbol));
        completed-nonterminals := add(completed-nonterminals, symbol);
        completion-items := concatenate(added-completions,
                                        completion-items);
        scan-items(added-completions);
      end if;
    end if;
  end for;
end method,
              

This section continues on page 28.

This section is referenced on page 27.

Completion items, as we mentioned, have their dots at the left end.

〈Local methods for completion-items 28〉+≡
method as-completion-item
    (production :: <production>)
 => (item :: <lr-item>);
  make-lr-item(automaton, production: production, dot: 0);
end method
              

Constructing the Canonical LR(0) Automaton

We will build the canonical LR(0) automaton when we initialize an instance of <lr-pars-
er-automaton>. As we build it, we will keep track of the states in the lr-parser-automa-
ton-itemsets table.

〈Slots for the <canonical-lr-parser-automaton> class28〉≡
constant slot lr-parser-automaton-itemsets :: <lr-itemset-table>
  = make(<lr-itemset-table>);
              

This section continues on page 30.

This section is referenced on page 23.

First we define the locate-lr-itemset method, which locates and returns the itemset for a kernel
set if it exists already, otherwise it constructs one. Note that we need to enter the itemset into the table
before we construct its successors, so that the recursive calls to locate-lr-itemset will find it.

〈Local methods for the <canonical-lr-parser-automaton> class initialize

method29〉≡
method locate-lr-itemset
    (kernel-items :: <list>)
 => (itemset :: <lr-itemset>);
  let itemset = element(automaton.lr-parser-automaton-itemsets,
                        kernel-items, default: #f);
  if (itemset)
    itemset;
  else
    let itemset
      = make-lr-itemset(automaton,
                        kernel-items: kernel-items);
    element(automaton.lr-parser-automaton-itemsets, kernel-items) := itemset;
    compute-reductions-and-successors!(itemset,
                                       completion-items(automaton,
                                                        grammar,
                                                        kernel-items));
    itemset;
  end if;
end method,
              

This section continues on page 29.

This section is referenced on page 23.

Transitions on terminal symbols occur when we shift. Transitions on nonterminals (gotos) occur after
a reduction. Though the successor itemsets are generated the same way, we will want to keep them
separate for later.

〈Additional slots for the <lr-itemset> class29〉≡
constant slot itemset-shift-table :: <object-table>
  = make(<object-table>);
constant slot itemset-goto-table :: <object-table>
  = make(<object-table>);
              

This section continues on pages 29 and 31.

This section is referenced on page 25.

We also collect the items with dot at the end, which are candidates for a reduction. Note that these
will also be in the kernel, except for items arising from empty productions (as noted in [1]).

〈Additional slots for the <lr-itemset> class 29〉+≡
slot itemset-reductions :: <list> = #();
              

To compute the successor states, we partition the kernel and completion items by the symbol after dot.

〈Local methods for the <canonical-lr-parser-automaton> class initialize method

29〉+≡
method compute-reductions-and-successors!
    (itemset :: <lr-itemset>, completions :: <list>)
 => ();
  let shift-kernels-table :: <object-table>
    = make(<object-table>);

  let goto-kernels-table :: <object-table>
    = make(<object-table>);
  local

    〈Define the local add-item-to-successor-kernel method 30〉;
  do(add-item-to-successor-kernel, itemset.itemset-kernel-items);
  do(add-item-to-successor-kernel, completions);

  for (kernel keyed-by symbol in shift-kernels-table)
    itemset.itemset-shift-table[symbol] := locate-lr-itemset(kernel);
  end for;

  for (kernel keyed-by symbol in goto-kernels-table)
    itemset.itemset-goto-table[symbol] := locate-lr-itemset(kernel);
  end for;
end method
              

Here we place the LR item into the right category. If dot is at the end, we could have a reduction
in this state. If the symbol after dot is a nonterminal, the item's successor (i.e., an item for the same
production with dot moved up one symbol) represents part of the parse state after a goto. If the symbol
is a terminal, the successor represents part of the parse state after a shift.

〈Define the local add-item-to-successor-kernel method30〉≡
method add-item-to-successor-kernel(item :: <lr-item>) => ();
  if (item-dot-at-end?(item))
    itemset.itemset-reductions := add(itemset.itemset-reductions, item);
  else
    let symbol = item-symbol-after-dot(item);
    if (grammar-symbol-nonterminal?(grammar, symbol))
      element(goto-kernels-table, symbol)
        := add(element(goto-kernels-table, symbol, default: #()),
               item-successor(item));
    else
      element(shift-kernels-table, symbol)
        := add(element(shift-kernels-table, symbol, default: #()),
               item-successor(item));
    end if;
  end if;
end method
              

This section is referenced on page 30.

To construct the start state, we augment the grammar with new productions, each deriving a user-spec-
ified start symbol followed by the end-of-input symbol. The item formed by putting dot at the left
end of such a production becomes the kernel of a start state. Calling locate-lr-itemset on this
kernel constructs the start state and the rest of the automaton.

〈Slots for the <canonical-lr-parser-automaton> class 28〉+≡
constant slot lr-parser-automaton-start-itemsets :: <object-table>
  = make(<object-table>);
              

(The exact value of unique-start-symbol is not important, as long as no grammar symbol ==
to it appears in the grammar.)

〈Construct the start state and its successors31〉≡
for(start-symbol in start-symbols)
  let unique-start-symbol = list(start-symbol);
  let start-production = make(<production>,
                              nonterminal: unique-start-symbol,
                              derives: vector(start-symbol, end-symbol));
  let start-item = make-lr-item(automaton,
                                production: start-production,
                                dot: 0);
  let start-state = locate-lr-itemset(list(start-item));
  automaton.lr-parser-automaton-start-itemsets[start-symbol] := start-state;

  〈Mark the accepting state 31〉
end for;
              

This section is referenced on page 24.

We can locate the accepting state by following the transitions on the start-symbol and end-
symbol.

〈Additional slots for the <lr-itemset> class 29〉+≡
slot itemset-accepting? :: <boolean> = #f;
              

〈Mark the accepting state31〉≡
let penultimate-state = start-state.itemset-goto-table[start-symbol];
let accepting-state = penultimate-state.itemset-shift-table[end-symbol];
accepting-state.itemset-accepting? := #t;
              

This section is referenced on page 31.

The initial state can be accessed using the public lr-parser-automaton-initial-state
method.

〈Module parser-automaton16〉+≡
define method lr-parser-automaton-initial-state
    (automaton :: <canonical-lr-parser-automaton>,
     start-symbol :: <object>)
 => (state :: <lr-itemset>)
  automaton.lr-parser-automaton-start-itemsets[start-symbol];
end method;
              

LR(0) Automata

Very few practical grammars are recognizable by an LR(0) automaton. However, for testing purposes
during development it was useful to define a concrete <lr0-parser-automaton> class.

〈Module parser-automaton16〉+≡
define class <lr0-parser-automaton> (<canonical-lr-parser-automaton>)
  // No additional slots
end class;
              

This demonstrates how we instantiate a concrete subclass of <parser-automaton>.

〈Module parser-automaton16〉+≡
define sealed method make-parser-automaton
    (class == #"LR-0", #rest key-value-pairs, #key, #all-keys)
 => (instance :: <lr0-parser-automaton>);
  apply(make, <lr0-parser-automaton>, key-value-pairs);
end method;
              

The <lr0-parser-automaton> class has associated with it concrete item and itemset classes
<lr0-item> and <lr0-itemset>, along with corresponding factory methods.

〈Module parser-automaton16〉+≡
define class <lr0-item> (<lr-item>)
  // No additional slots
end class;
              
define method make-lr-item
    (automaton :: <lr0-parser-automaton>,
     #rest key-value-pairs, #key, #all-keys)
 => (instance :: <lr-item>);
  apply(make, <lr0-item>, key-value-pairs);
end method;
              
define class <lr0-itemset> (<lr-itemset>)
  // No additional slots
end class;
              
define method make-lr-itemset
    (automaton :: <lr0-parser-automaton>,
     #rest key-value-pairs, #key, #all-keys)
 => (instance :: <lr-itemset>);
  apply(make, <lr0-itemset>, key-value-pairs);
end method;
              

Since the canonical LR automaton is the LR(0) automaton, there's not much to do on initialization
but verify that the result really is a valid LR(0) automaton. With no lookahead, then we have a re-
duce/reduce conflict in any itemset with more than one reduce action, and a shift/reduce conflict in
any itemset with both shift actions and reduce actions.

〈Module parser-automaton16〉+≡
define method initialize
    (automaton :: <lr0-parser-automaton>,
     #next next-method,
     #key, #all-keys)
 => ();
  next-method();
  for(itemset in automaton.lr-parser-automaton-itemsets)
    if(itemset.itemset-reductions.size > 1)
      error(make(<parser-automaton-reduce/reduce-error>,
                 productions: itemset.itemset-reductions));
    elseif(~empty?(itemset.itemset-reductions)
           & ~empty?(itemset.itemset-shift-table))
      error(make(<parser-automaton-shift/reduce-error>,
            productions: reduce(concatenate,
                                itemset.itemset-reductions,
                                map(itemset-kernel-items,

                                    itemset.itemset-shift-table))));
    end if;
  end for;
end method;
              

The following methods determine what actions an LR(0) parser should take with a given state and
input, using our <lr-itemset> representation.

〈Module parser-automaton16〉+≡
define method lr-parser-automaton-terminal-transition
    (automaton :: <canonical-lr-parser-automaton>,
     state :: <lr-itemset>,
     terminal :: <object>)
 => (action :: one-of(#"shift", #"reduce", #"error", #"accept"),
     data :: <object>);
  let dest = element(state.itemset-shift-table, terminal, default: #f);
  if(dest)
    if(itemset-accepting?(dest))
      values(#"accept", #t)
    else
      values(#"shift", dest);
    end if;
  else
    if(~empty?(state.itemset-reductions))
      values(#"reduce", first(state.itemset-reductions));
    else
      values(#"error", #f);
    end if;
  end if;
end method;
              
define method lr-parser-automaton-nonterminal-transition
    (automaton :: <canonical-lr-parser-automaton>,
     state :: <lr-itemset>,
     nonterminal :: <object>)
 => (new-state :: <object>);
  state.itemset-goto-table[nonterminal];
end method;
              

With no lookahead, the list of legal transitioning terminals is not available in states containing reduc-
tions for LR(0).

〈Module parser-automaton16〉+≡
define method lr-parser-automaton-transition-terminals
    (automaton :: <canonical-lr-parser-automaton>,
     state :: <lr-itemset>)
 => (terminals :: false-or(<sequence>))
  let terminals = state.itemset-shift-table.key-sequence;
  if(empty?(terminals))
    #f;
  else
    terminals;
  end if;
end method;

〈Module parser-automaton16〉+≡
define method lr-parser-automaton-transition-nonterminals
    (automaton :: <canonical-lr-parser-automaton>,
     state :: <lr-itemset>)
 => (nonterminals :: false-or(<sequence>))
  let nonterminals = state.itemset-goto-table.key-sequence;
  if(empty?(nonterminals))
    #f;
  else
    nonterminals;
  end if;
end method;
              

Generating an SLR(1) Automaton from a Grammar

The set of SLR(1) grammars is a strict subset of the set of LALR(1) grammars, and there are many
programming language constructs that can be recognized by a LALR(1) automaton but not an SLR(1)
automaton. Also, the parser tables for SLR(1) automata are not significantly different in size from
those for LALR(1) automata, so there is no compelling practical reason to use SLR(1) over LALR(1).
However, the SLR(1) lookahead sets are easier to generate than LALR(1) lookahead sets, so we'll use
the SLR(1) lookahead generation code as a preview of the sort of issues and techniques involved with
LALR(1) lookahead generation.

〈Module parser-automaton16〉+≡
define class <slr1-parser-automaton> (<canonical-lr-parser-automaton>)
  // No additional slots
end class;
            
define sealed method make-parser-automaton
    (class == #"SLR-1", #rest key-value-pairs, #key, #all-keys)
 => (instance :: <slr1-parser-automaton>);
  apply(make, <slr1-parser-automaton>, key-value-pairs);
end method;
            

If we add lookahead context information to the items that could potentially be reduced, we can elim-
inate some conflicts. In particular, for LR(1) family parsers, we determine what terminal symbols
might potentially be shifted after the reduction of the item, and perform the appropriate reduction
when we see one of its lookahead symbols.

〈Module parser-automaton16〉+≡
define class <lr1-item> (<lr-item>)
  slot item-lookahead :: <list>;
end class;
            
define method make-lr-item
    (automaton :: <slr1-parser-automaton>,
     #rest key-value-pairs, #key, #all-keys)
 => (instance :: <lr-item>);
  apply(make, <lr1-item>, key-value-pairs);
end method;

Since (in states that require lookahead) we will only reduce an item when one of the item's lookahead
symbols is seen, each itemset needs to have a table mapping input terminals to the productions to
be reduced.

〈Module parser-automaton16〉+≡
define class <lr1-itemset> (<lr-itemset>)
  constant slot itemset-reduction-table :: <object-table>
    = make(<object-table>);
end class;
            
define method make-lr-itemset
    (automaton :: <slr1-parser-automaton>,
     #rest key-value-pairs, #key, #all-keys)
 => (instance :: <lr-itemset>);
  apply(make, <lr1-itemset>, key-value-pairs);
end method;
            

SLR(1) associates a one-terminal lookahead set with each item that has its dot at the end, just as the
other LR(1) family parser generation algorithms do. The FOLLOW set of each item's nonterminal,
consisting of all of the terminal symbols that could follow as the next terminal to be shifted after a
derivation of the nonterminal, is used as the lookahead set for SLR(1) parsers.

The FOLLOW sets of the nonterminals are not a property of the LR automaton per se, but are prop-
erties of the grammar itself. However, after we have built an LR(0) automaton, computing FOLLOW
sets does become computationally less expensive, as we'll see..

〈Module parser-automaton16〉+≡
define method initialize
    (automaton :: <slr1-parser-automaton>,
     #next next-method,
     #key grammar :: <grammar>,
          start-symbol :: <object>,
          end-symbol :: <object>,
          full-lookahead? :: <object> = #f,
     #all-keys)
 => ();
  next-method();

  〈Compute SLR(1) lookaheads for automaton 39〉
end method;
            

Terminal symbols in a nonterminal's FOLLOW set can come from two different sources (see [8]):

■ If the nonterminal appears in a production with symbols following it on the right, then the FIRST
set of that sequence of symbols is included in the nonterminal's FOLLOW set.

■ If the nonterminal appears at the end of a production, or all of the symbols appearing after it are
nullable, then the FOLLOW set of that production's left-hand side nonterminal is included in the
nonterminal's FOLLOW set.

For the first case, we can compute the symbols following the nonterminal by looking at all of the
goto-transitions on the nonterminal, and computing what terminal symbols could appear in each des-
tination itemset. Kristensen and Madsen [13] call this the TRANS set of the destination itemset; for
lack of a better name, we will do the same. The portion of the FOLLOW set for a nonterminal due

to this case, then, is the union of all of the TRANS sets of itemsets that follow some transition on
the nonterminal.

For the second case, we need to determine for each nonterminal what other nonterminals have a right-
dependency on it, and therefore contribute to its FOLLOW set.

Computing TRANS Sets

Kristensen and Madsen show in their lemma 5.2 that the TRANS set of an itemset is equivalent to
the union over all of the itemset's kernel items of the FIRST sets of the portion after each item's dot.
We can compute this easily enough.

At the same time, we can identify some of the nonterminals that have a right-dependency on the
nonterminal symbol before dot, since this occurs when all of the symbols after dot are nullable.

〈Module parser-automaton16〉+≡
define method itemset-trans-symbols
    (grammar :: <grammar>,
     itemset :: <lr1-itemset>)
 => (trans-symbols :: <sequence>,
     right-dependency-items :: <sequence>);
  local

    〈Local itemset-trans-symbols methods 36〉;

  let trans-symbols = #();
  let right-dependency-items = #();
  for(item in itemset.itemset-kernel-items)
    let (first-symbols, nullable?)
      = production-first-symbols-after(item.item-production, item.item-dot);
    trans-symbols := union(trans-symbols, first-symbols);
    if(nullable?)
      right-dependency-items
        := add(right-dependency-items, item);
    end if;
  end for;
  values(trans-symbols, right-dependency-items);
end method;
              

〈Local itemset-trans-symbols methods36〉≡
method production-first-symbols-after
    (production :: <production>, start :: <integer>)
 => (first-symbols-after :: <sequence>,
     nullable-after? :: <boolean>);
  for (index from start below production.production-derives.size,
       first-symbols = #()
         then union(grammar-symbol-first-symbols
                      (grammar, production.production-derives[index]),
                    first-symbols),
       prefix? = #t
         then grammar-symbol-nullable?
               (grammar, production.production-derives[index]),
       while: prefix?)
  finally
    values(first-symbols, prefix?);

  end for;
end
              

This section is referenced on page 36.

Dependent Sets

The SLR(1) FOLLOW set is an instance of a set computed from an initial value (in this case the union
of the TRANS sets) and the union of the values of certain other sets that it depends on (for SLR(1), the
sets of nonterminals that have a right-dependency on the current one). Since LALR(1) computation
will also involve sets of this kind, we'll define a corresponding abstraction, <dependent-set>.

〈Exports for the grammar library 1〉+≡
export dependent-set;
              

〈Utility module definitions for the grammar library 1〉+≡
define module dependent-set
  use common-dylan;
  use digraph;
  export <dependent-set>,
         dependent-set-initial,
         dependent-set-initial-setter,
         dependent-set-value,
         dependent-set-dependents,
         add-set-dependent!;
end module;
              

The class itself has a dependent-set-initial slot for the initial component and a (non-export-
ed) dependent-set-final-value slot for memoizing the computed set.

〈Module dependent-set 37〉≡
define class <dependent-set> (<object>)
  slot dependent-set-initial :: <sequence> = #(),
    init-keyword: initial:;
  slot dependent-set-final-value :: <sequence>;
  slot dependent-set-dependents :: <list> = #();
end class;
              
define method add-set-dependent!
    (set :: <dependent-set>, depends-on :: <dependent-set>)
 => ();
  set.dependent-set-dependents
    := add-new!(set.dependent-set-dependents, depends-on);
end method;
              

This module continues on page 37.

There might be a circular dependency among <dependent-set> objects. So, to compute the union
of a set's initial value and the final values of the sets it depends on, we need to use the find-root-
ed-scc function to find the fixed-point solution.

〈Module dependent-set37〉+≡
define method dependent-set-value
    (set :: <dependent-set>)

 => (sequence :: <sequence>);
  if(~slot-initialized?(set, dependent-set-final-value))
    local

      〈Define local visit and collapse methods for dependent-set-value 38〉;
    find-rooted-scc(set, visit, collapse);
  end if;
  set.dependent-set-final-value;
end method;
              

We start with the initial value, and union it with the values of each of the other sets.

〈Define local visit and collapse methods for dependent-set-value38〉≡
method visit
    (set :: <dependent-set>, traverse :: <function>)
 => ();
  set.dependent-set-final-value := set.dependent-set-initial;

  for(dependent in set.dependent-set-dependents,
      final-value = set.dependent-set-initial
        then
          begin
            if(~slot-initialized?(dependent, dependent-set-final-value))
              traverse(dependent);
            end if;
            union(final-value, dependent.dependent-set-final-value);
          end)
  finally
    set.dependent-set-final-value := final-value;
  end for;
end method,
              

This section continues on page 38.

This section is referenced on page 38.

Mutually recursive dependent-sets share the same value.

〈Define local visit and collapse methods for dependent-set-value 38〉+≡
method collapse
    (scc-head :: <dependent-set>, scc-other :: <dependent-set>)
 => ();
  scc-other.dependent-set-final-value := scc-head.dependent-set-final-value;
end method
              

Computing SLR(1) Lookahead Sets

During their construction, we'll represent FOLLOW sets as <dependent-set> objects.

〈Modules imported by the parser-automaton module38〉≡
use dependent-set;
              

This section is referenced on page 14.

We'll use an <object-table> with one entry for each nonterminal to store the FOLLOW sets.

〈Compute SLR(1) lookaheads for automaton39〉≡

let follow = make(<object-table>);
              
local
  method follow-dependent-set(nonterminal)
   => (set :: <dependent-set>);
    element(follow, nonterminal, default: #f)
      | (follow[nonterminal] := make(<dependent-set>, initial: #()));
  end;
              

This section continues on pages 39 and 39.

This section is referenced on page 35.

Now we go through each state in the LR(0) automaton and compute the TRANS sets of the successor
states of nonterminals. We also keep track of the right-dependency relationship for the FOLLOW
computation.

〈Compute SLR(1) lookaheads for automaton 39〉+≡
for(itemset in automaton.lr-parser-automaton-itemsets)
  for(goto-itemset keyed-by goto-symbol in itemset.itemset-goto-table)
    let (trans-symbols, right-dependency-items)
      = itemset-trans-symbols(grammar, goto-itemset);
    let set = follow-dependent-set(goto-symbol);
    set.dependent-set-initial
      := union(set.dependent-set-initial, trans-symbols);
    for(item in right-dependency-items)
      let rd = item.item-production.production-nonterminal;
      add-set-dependent!(set, follow-dependent-set(rd));
    end for;
  end for;
end for;
              

Once the initial value and dependencies have been collected, we can use dependent-set-value
to find the FOLLOW set for each nonterminal. We'll use this value to set the item-lookahead of
all of the items where a reduction could potentially occur, and place entries in the itemset's item-
set-reduction-table. (We don't bother to compute the lookahead set in states where there is
no potential ambiguity without lookahead, unless full-lookahead? is true.

〈Compute SLR(1) lookaheads for automaton 39〉+≡
for(itemset in automaton.lr-parser-automaton-itemsets)
  if(full-lookahead?
     | (itemset.itemset-reductions.size > 1)
     | (itemset.itemset-reductions.size > 0
        & itemset.itemset-shift-table.size > 0))
    for(item in itemset.itemset-reductions)
      let nonterminal = item.item-production.production-nonterminal;
      item.item-lookahead
        := follow-dependent-set(nonterminal).dependent-set-value;
      fill-itemset-reduction-table!(itemset, item);
    end for;
  end if;
end for;

As we add entries to the itemset-reduction-table, we check for other reductions with the
same lookahead symbol (reduce/reduce conflicts) or shift actions transitioning on one of our looka-
head symbols (shift/reduce conflicts).

〈Module parser-automaton16〉+≡
define method fill-itemset-reduction-table!
    (itemset :: <lr1-itemset>, item :: <lr1-item>)
 => ();
  for(lookahead-symbol in item.item-lookahead)
    let production = element(itemset.itemset-reduction-table,
                             lookahead-symbol, default: #f);
    if(production)

       〈Signal a reduce/reduce error 40〉
    else
      let shift-itemset = element(itemset.itemset-shift-table,
                                  lookahead-symbol, default: #f);
      if(shift-itemset)

         〈Signal a shift/reduce error 40〉
      else
        element(itemset.itemset-reduction-table, lookahead-symbol)
          := item.item-production;
      end if;
    end if;
  end for;
end method;
              

When signalling a <parser-automaton-reduce/reduce-error>, we also establish a han-
dler for <parser-automaton-shift/reduce-restart> to allow callers to supply their
own rules for disambiguating items.

〈Signal a reduce/reduce error40〉≡
block ()
  error(make(<parser-automaton-reduce/reduce-error>,
             productions: list(item.item-production, production),
             inputs: list(lookahead-symbol)));
exception (restart :: <parser-automaton-reduce/reduce-restart>)
  if (restart.parser-automaton-restart-action == item.item-production)
    element(itemset.itemset-reduction-table, lookahead-symbol)
      := item.item-production;
  elseif (restart.parser-automaton-restart-action ~== production)
    error("bad recovery action %=", restart.parser-automaton-restart-action);
  end if;
end block;
              

This section is referenced on page 40.

We do the same for <parser-automaton-shift/reduce-error>..

〈Signal a shift/reduce error40〉≡
block ()
  error(make(<parser-automaton-shift/reduce-error>,
             productions:
               pair(item.item-production,
                    map-as(<list>, item-production,
                           shift-itemset.itemset-kernel-items)),

             inputs: list(lookahead-symbol)));
exception (restart :: <parser-automaton-shift/reduce-restart>)
  if (restart.parser-automaton-restart-action == item.item-production)
    remove-key!(itemset.itemset-shift-table, lookahead-symbol);
    element(itemset.itemset-reduction-table, lookahead-symbol)
      := item.item-production;
  elseif (restart.parser-automaton-restart-action ~== #"shift")
    error("bad recovery action %=", restart.parser-automaton-restart-action);
  end if;
end block;
              

This section is referenced on page 40.

Querying the Automaton

The following methods determine what actions a parser should take with a given state and input,
using the <lr1-itemset> representation. Lookahead information allows us to determine which
reduction to take in states that would be ambiguous without it. In states where no lookahead is required,
and we can't shift on the given input, we perform the default reduction if there is one.

〈Module parser-automaton16〉+≡
define method lr-parser-automaton-terminal-transition
    (automaton :: <canonical-lr-parser-automaton>,
     state :: <lr1-itemset>,
     terminal :: <object>)
 => (action :: one-of(#"shift", #"reduce", #"error", #"accept"),
     data :: <object>);
  let dest = element(state.itemset-shift-table, terminal, default: #f);
  if(dest)
    if(itemset-accepting?(dest))
      values(#"accept", #t)
    else
      values(#"shift", dest);
    end if;
  else
    let reduce = element(state.itemset-reduction-table, terminal, default: #f);
    if(reduce)
      values(#"reduce", reduce);
    elseif(state.itemset-reductions.size = 1)
      values(#"reduce", first(state.itemset-reductions).item-production);
    else
      values(#"error", #f);
    end if;
  end if;
end method;
              
define method lr-parser-automaton-nonterminal-transition
    (automaton :: <canonical-lr-parser-automaton>,
     state :: <lr1-itemset>,
     nonterminal :: <object>)
 => (new-state :: <object>);
  state.itemset-goto-table[nonterminal];
end method;

The set of terminals that could transition of of a state includes the set of terminals that would cause
the parser to shift, and the set of lookahead terminal symbols that would allow a reduction.

〈Module parser-automaton16〉+≡
define method lr-parser-automaton-transition-terminals
    (automaton :: <canonical-lr-parser-automaton>,
     state :: <lr1-itemset>)
 => (terminals :: false-or(<sequence>))
  // FIXME Make sure this is complete
  concatenate(state.itemset-shift-table.key-sequence,
              state.itemset-reduction-table.key-sequence);
end method;
              

Generating an LALR(1) Automaton from a Grammar

When DeRemer [6][7] first proposed the SLR(k) and LALR(k) subsets of LR(k), he gave practical
methods for generating SLR(k) parsers only. (He proposed building the LALR(k) automaton by build-
ing an LR(k) automaton, and merging itemsets that are equivalent except for their lookahead sets.)
Several others subsequently developed algorithms for generating LALR(1) or LALR(k) automata,
both efficient and otherwise. The algorithm used by the yacc program, while not the most efficient,
has received the widest exposure in [1]. Other algorithms related to the one we'll be using are outlined
in [13], [8], [14], [11], [15], and [12].

More specifically, we'll be using Ives' algorithm [12], which is basically an application of the Eve
and Kurki-Suonio transitive closure algorithm ([9], our find-rooted-scc) to Kristensen and
Madsen's algorithm [13], further modifying it to use kernel items only. (Got that?)

Many of these references are somewhat bogged down in notation, and each one uses a slightly different
set of notational conventions. To avoid confusing things, we'll do without any mathematical notation
other than our program code.

〈Module parser-automaton16〉+≡
define class <lalr1-parser-automaton> (<canonical-lr-parser-automaton>)
  // No additional slots
end class;
            
define sealed method make-parser-automaton
    (class == #"LALR-1", #rest key-value-pairs, #key, #all-keys)
 => (instance :: <lalr1-parser-automaton>);
  apply(make, <lalr1-parser-automaton>, key-value-pairs);
end method;
            

LALR(1) automata will use the same item and itemset classes as SLR(1) automata.

〈Module parser-automaton16〉+≡
define method make-lr-item
    (automaton :: <lalr1-parser-automaton>,
     #rest key-value-pairs, #key, #all-keys)
 => (instance :: <lr-item>);
  apply(make, <lr1-item>, key-value-pairs);
end method;

define method make-lr-itemset
    (automaton :: <lalr1-parser-automaton>,
     #rest key-value-pairs, #key, #all-keys)
 => (instance :: <lr-itemset>);
  apply(make, <lr1-itemset>, key-value-pairs);
end method;
             

〈Module parser-automaton16〉+≡
define method initialize
    (automaton :: <lalr1-parser-automaton>,
     #next next-method,
     #key grammar :: <grammar>,
          start-symbol :: <object>,
          end-symbol :: <object>,
          full-lookahead? :: <object> = #f,
     #all-keys)
 => ();
  next-method();

  〈Compute LALR(1) lookaheads for automaton 43〉
end method;
            

When we were computing SLR(1) lookaheads, we lumped together the TRANS sets of all itemsets
following a given nonterminal. For LALR(1) we will keep track of more fine-grained lookahead in-
formation—one set for each kernel item, in fact.

〈Compute LALR(1) lookaheads for automaton43〉≡
let lookahead = make(<object-table>);
local
  method lookahead-dependent-set
      (item :: <lr-item>)
   => (set :: <dependent-set>);
    element(lookahead, item, default: #f)
      | (lookahead[item] := make(<dependent-set>, initial: #()));
  end method;
            

This section continues on pages 44, 46, 46, 47, 47, 48 and 49.

This section is referenced on page 43.

How do the lookahead sets get associated with items? Consider the sequence of events that happens
when a production is reduced:

1. Elements associated with the right-hand side of the production are popped off the stack, returning
to the state where items for that production were first introduced.

2. The nonterminal that was reduced is pushed onto the stack.
3. The goto-transition on the nonterminal is taken from this state.
4. The TRANS set of the new set contains some of the terminals that might appear at this point. A

reduction at this point might allow additional terminals to appear.
We could get the TRANS set we want by tracing transitions backwards until we reach the state con-
taining the goto-transition. We haven't really got the data structures for tracing backwards, however;
it's much more natural to start from states containing goto-transitions, and add the elements of the
TRANS set to items we reach by tracing through to successor states.

That's not the whole of it, however. As we noted above, additional productions might be reduced
without any additional terminal symbols being shifted. This means that the LALR lookahead set for
an item also depends on the lookahead sets of items that might be reduced immediately afterwards,
just as the FOLLOW set of a nonterminal depends on the FOLLOW sets of nonterminals that right-
depend on it.

A naïve approach to determining possible lookaheads would be to define a “lookahead” set for each of
the propagating (FIXME terminology) items in the set after the goto-transition, making it depend in-
directly on the lookahead of the successor items where the reduction might actually occur. The prob-
lem with this method is that more than one state might have a goto-transition to this state, thus losing
context information and making the lookahead information more coarse-grained. In fact, this results
in a so-called NQLALR(1), or “Not Quite” LALR(1) lookahead set. (See [8] and [10].) NQLALR(1)
automata are less powerful than LALR(1) automata, and (as demonstrated in [3]) may not necessarily
be any better than SLR(1).

Instead, we define lookaheads as they are normally defined for LR(1) automata (see [1]). A terminal
symbol is included in an item's lookahead set if the terminal could appear immediately after the item's
production is reduced, provided the automaton passed through the item's state before doing so. This
defines a lookahead set for every item. Not every item needs one, however, unless either

■ It is a final item that might be reduced; or

■ It propagates its lookaheads to other items.
If an item has dot in front of a nonterminal, and all of the symbols after the nonterminal are nullable
(meaning that the item's production could be reduced without shifting any more terminals), then the
item will propagate its lookaheads to items generated from the nonterminal's productions (which might
potentially be reduced immediately before).

Some of these propagating items might be completion items, which we don't store explicitly in our
<lr-parser-automaton> itemsets. To solve this problem, we can determine beforehand which
nonterminals cause propagating items to appear in completion sets, and propagate lookaheads to the
appropriate items from the generating kernel item.

An Alternate Formulation for Computing TRANS Sets

We can pre-compute TRANS sets and establish the lookahead set dependencies at the same time by
using an alternative formulation of the definition of a TRANS set. In particular, an itemset's TRANS
set is composed of the terminals that could appear immediately without further reductions (the keys
in the itemset's itemset-shift-table), unioned with the TRANS sets of itemsets that follow
from nullable nonterminals.

Our implementation here memoizes both the TRANS dependent-set and the list of right-dependent
items for each itemset.

〈Compute LALR(1) lookaheads for automaton 43〉+≡
let trans = make(<object-table>);
let right-dep = make(<object-table>);
local
  method trans-dependent-set
      (itemset :: <lr1-itemset>)
   => (trans-set :: <dependent-set>,
       right-dependency-items :: <sequence>);

    let trans-set = element(trans, itemset, default: #f);
    if(trans-set)
      values(trans-set, right-dep[itemset]);
    else
      let trans-set
        = make(<dependent-set>,
               initial: as(<list>,
                           key-sequence(itemset.itemset-shift-table)));
      for(goto-itemset keyed-by nonterminal in itemset.itemset-goto-table)
        if(grammar-symbol-nullable?(grammar, nonterminal))
          let (goto-trans, goto-rd) = trans-dependent-set(goto-itemset);
          add-set-dependent!(trans-set, goto-trans);
        end if;
      end for;

      〈Compute right-dependency-items 45〉
      trans[itemset] := trans-set;
      right-dep[itemset] := right-dependency-items;
      values(trans-set, right-dependency-items);
    end if;
  end method;
              

An item is in right-dependency-items either if dot is at the end, or if dot is before a nul-
lable nonterminal and the destination of the goto-transition has one or more right-dependen-
cy-items.

〈Compute right-dependency-items45〉≡
let right-dependency-items = #();
for(item in itemset.itemset-kernel-items)
  if(item-dot-at-end?(item))
    right-dependency-items := add(right-dependency-items, item);
  else
    let sym = item-symbol-after-dot(item);
    if(grammar-symbol-nullable?(grammar, sym))
      let goto-itemset = itemset.itemset-goto-table[sym];
      if(~empty?(right-dep[goto-itemset]))
        right-dependency-items := add(right-dependency-items, item);
      end if;
    end if;
  end if;
end for;
              

This section is referenced on page 45.

Preparing for the LALR(1) Lookahead Computation

Now we can do three things at once with a pass through all of the itemsets that are destinations of
goto-transitions:

■ Pre-compute the TRANS set for each such itemset.

■ Determine the earliest point at which a given production will propagate its lookaheads. This is one
less than the earliest dot where one of the production's items appears in a right-dependen-
cy-items list (implying that all of the symbols after dot are nullable).

We do this because only final items and propagating items actually need to have lookahead sets
associated with them.

■ Determine which nonterminals cause propagating items to appear in completion sets. This happens
when the derivation of a nonterminal is another nonterminal followed by zero or more nullable
symbols. This makes the nonterminals “equivalent” in some sense.

We'll use a <dependent-set> to note equvalence relationships between nonterminals. A nonter-
minal is always a member of its own equiv-dependent-set.

〈Compute LALR(1) lookaheads for automaton 43〉+≡
let equiv-nonterminals = make(<object-table>);
local
  method equiv-dependent-set
      (nonterminal :: <object>)
   => (set :: <dependent-set>);
    element(equiv-nonterminals, nonterminal, default: #f)
      | (equiv-nonterminals[nonterminal]
           := make(<dependent-set>, initial: list(nonterminal)));
  end method;
              

Our pass through all possible destinations of goto-transitions:

〈Compute LALR(1) lookaheads for automaton 43〉+≡
let propagates-dot = make(<object-table>);
for(itemset in automaton.lr-parser-automaton-itemsets)
  for(goto-itemset keyed-by goto-symbol in itemset.itemset-goto-table)
    unless(element(trans, goto-itemset, default: #f))
      let (trans-set, right-dependency-items)
        = trans-dependent-set(goto-itemset);
      for(rd-item in right-dependency-items)
        if(rd-item.item-dot = 1)

          〈Note that rd-item.item-production.production-nonterminal can derive goto-symbol 46〉
        else

          〈See if this is the smallest dot where rd-item propagates 46〉
        end if;
      end for;
    end unless;
  end for;
end for;
              

〈Note that rd-item.item-production.production-nonterminal can derive go-

to-symbol46〉≡
add-set-dependent!
  (equiv-dependent-set(rd-item.item-production.production-nonterminal),
   equiv-dependent-set(goto-symbol));
              

This section is referenced on page 46.

〈See if this is the smallest dot where rd-item propagates46〉≡
unless(element(propagates-dot, rd-item.item-production, default: #f)
       & propagates-dot[rd-item.item-production] < rd-item.item-dot - 1)
  propagates-dot[rd-item.item-production] := rd-item.item-dot - 1;
end unless;

This section is referenced on page 46.

Computing LALR(1) Lookahead Sets

We need to “contribute” sets containing lookahead symbols to the items to which that they apply.
We follow transitions to find the destination itemsets containing the items, stopping at the first item
that propagates.

Given a starting state and a sequence of symbols, the goto method computes the destination state
after transitioning on each of the symbols until we reach the item whose dot equals end-dot.

〈Compute LALR(1) lookaheads for automaton 43〉+≡
local
  method goto
      (itemset :: <lr1-itemset>, symbols :: <sequence>,
       start-dot :: <integer>, end-dot :: <integer>)
   => (destination :: <lr1-itemset>);
    if(start-dot == end-dot)
      itemset;
    else
      let symbol = symbols[start-dot];
      let goto-itemset
        = element(itemset.itemset-goto-table, symbol, default: #f);
      if(goto-itemset)
        goto(goto-itemset, symbols, start-dot + 1, end-dot);
      else
        goto(itemset.itemset-shift-table[symbol], symbols,
             start-dot + 1, end-dot);
      end if;
    end if;
  end method,
              

The contribute! method adds set containing lookaheads from a TRANS set or a propagating
item to the relevant items. The items for empty productions are not found in the kernel, but are in the
list of productions ready for reduction.

〈Compute LALR(1) lookaheads for automaton 43〉+≡
  method contribute!
      (itemset :: <lr1-itemset>, nonterminal :: <object>,
       set :: <dependent-set>)
   => ();
    for(production in grammar-symbol-productions(grammar, nonterminal))
      block(done)
        for(kernel-item in if(empty?(production.production-derives))
                             itemset.itemset-reductions
                           else
                             let min-dot
                               = element(propagates-dot, production,
                                         default:
                                           production.production-derives.size);
                             let dest
                               = goto(itemset, production.production-derives,
                                      0, min-dot);
                             dest.itemset-kernel-items

                           end if)
          if(kernel-item.item-production == production)
            add-set-dependent!(lookahead-dependent-set(kernel-item), set);
            done();
          end if;
        end for;
      end block;
    end for;
  end method;
              

Here's where we propagate the TRANS set and propagated lookaheads to the items that are ready for
reduction or propagating items that precede them.

〈Compute LALR(1) lookaheads for automaton 43〉+≡
for(itemset in automaton.lr-parser-automaton-itemsets)

  〈Contribute TRANS sets to final and propagating items 48〉

  〈Contribute propagated lookaheads to final and propagating items 48〉
end for;
              

We contribute TRANS set information to items that might have been reduced immediately before
transitioning on each nonterminal, or to propagating items that precede them.

〈Contribute TRANS sets to final and propagating items48〉≡
for(goto-itemset keyed-by goto-symbol in itemset.itemset-goto-table)
  let (trans-set, right-dependency-items)
    = trans-dependent-set(goto-itemset);
  unless(empty?(trans-set.dependent-set-value))
    for(nonterminal in equiv-dependent-set(goto-symbol).dependent-set-value)
      contribute!(itemset, nonterminal, trans-set);
    end for;
  end unless;
end for;
              

This section is referenced on page 48.

Here we contribute the lookahead sets of items that propagate to items that might have been reduced
immediately before.

〈Contribute propagated lookaheads to final and propagating items48〉≡
for(kernel-item in itemset.itemset-kernel-items)
  unless(item-dot-at-end?(kernel-item))
    let prop = element(propagates-dot, kernel-item.item-production,
                       default: #f);
    if(prop & prop <= kernel-item.item-dot)
      let dot-symbol = item-symbol-after-dot(kernel-item);
      let my-lookahead = lookahead-dependent-set(kernel-item);
      for(nonterminal in equiv-dependent-set(dot-symbol).dependent-set-value)
        contribute!(itemset, nonterminal, my-lookahead);
      end for;

      〈Let the successor of propagating item kernel-item depend on it 49〉
    end if;
  end unless;
end for;

This section is referenced on page 48.

Note that if we stop contributing lookaheads at the first propagating item, then successor items won't
have any lookahead set associated with them. Since the successor of an item will have all of the
same lookahead symbols, we add the lookahead set of a propagating item to its successor using the
dependence relationship.

〈Let the successor of propagating item kernel-item depend on it49〉≡
let dest = itemset.itemset-goto-table[dot-symbol];
block(done)
  for(dest-item in dest.itemset-kernel-items)
    if(dest-item.item-production == kernel-item.item-production)
      add-set-dependent!(lookahead-dependent-set(dest-item), my-lookahead);
      done();
    end if;
  end for;
end block;
              

This section is referenced on page 48.

Just as with the SLR(1) lookahead sets, once the initial value and dependencies have been collected, we
can use dependent-set-value to find the lookahead set for each nonterminal, and use fill-
itemset-reduction-table! to fill in tables and check for conflicts.

〈Compute LALR(1) lookaheads for automaton 43〉+≡
for(itemset in automaton.lr-parser-automaton-itemsets)
  if(full-lookahead?
     | (itemset.itemset-reductions.size > 1)
     | (itemset.itemset-reductions.size > 0
        & itemset.itemset-shift-table.size > 0))
    for(item in itemset.itemset-reductions)
      let nonterminal = item.item-production.production-nonterminal;
      item.item-lookahead
        := lookahead-dependent-set(item).dependent-set-value;
      fill-itemset-reduction-table!(itemset, item);
    end for;
  end if;
end for;

